UNIVERSIDADE FEDERAL DE ALAGOAS

INSTITUTO DE COMPUTAÇÃO

COMPILADORES 2021.1

Linguagem JORG

Gramática LL(1)

Gabriel Luiz Leite Souza

João Victor Falcão Santos Lima

Rodrigo Santos da Silva

Maceió

2021

```
S = DecFun S | DecVar S | ε

DecFun = 'function' TipoFunc NomeFunc '(' ParamsDec ')' Bloco

TipoFunc = Tipo | 'void'

Tipo = 'int' | 'float' | 'char' | 'string' | 'bool'

NomeFun = 'id' | 'main'

ParamsDec = Tipo 'id' Arr ParamsDecX | ε

ParamsDecX = ',' ParamsDec | ε

Arr = '[' ']' | ε

DecVar = Tipo ListId ';' | 'const' Tipo ListId ';'

ListId = AtribOuId ListIdX

ListIdX = ',' ListId | ε

AtribOuId = 'id' ArrId Atrib

ArrId = '[' Earit ']' | ε

Atrib = '=' EconcOuListArr | ε

EconcOuListArr = Econc | '[' ListArr ']'

ListArr = Econc ListArrX

ListArrX = ',' ListArr | ε

ExpAtrib = ArrId Atrib ‘;’
```

```
ChamadaFunc = '(' ParamsChamada ')' ';'

ParamsChamada = Econc ParamsChamadaX | ε

ParamsChamadaX = ',' ParamsChamada | ε

Bloco = '{' Sentencas '}'

Sentencas = DecVar Sentencas | Cmd Sentencas
               | ChamadaFuncOuAtrib Sentencas | ε

ChamadaFuncOuAtrib = ChamadaFunc | ExpAtrib

Cmd = ‘return’ Ret ';'

Ret = Econc | ε

Cmd = ‘write’ '(' ParamsChamada ')' ';'

Cmd = ‘writeln’ '(' ParamsChamada ')' ';'

Cmd = input '(' LId ')' ';'

LId = Id LIdX

LIdX = ',' LId | ε

Id = ‘id’ ArrId

Cmd = ‘if’ '(' Ebool ')' Bloco Else

Else = ‘else’ Bloco | ε

Cmd = ‘while’ '(' Ebool ')' Bloco

Cmd = ‘break’ ';'

Cmd = ‘for’ '(' Int ',' Int ',' Int Incr ')' Bloco
```

```
Int = 'id' | ‘int’ 'id' | 'CONST_INT'

Incr = ',' Int | ε

Econc = Ebool EconcX

EconcX = ‘OP_CONCAT’ Ebool EconcX | ε

Ebool = Tbool EboolX

EboolX = ‘OP_OR’ Tbool EboolX | ε

Tbool = Fbool TboolX

TboolX = ‘OP_AND’ Fbool TboolX | ε

Fbool = ‘OP_NOT’ Trelac | Trelac

Trelac = Earit TrelacX

TrelacX = OpRelac Earit TrelacX | ε

Earit = Tarit EaritX

EaritX = OpArit Tarit EaritX | ε

Tarit = Parit TaritX

TaritX = OpMult Parit TaritX | ε

Parit = Farit ParitX

ParitX = OpPot Parit | ε

Farit = OpArit FaritX | FaritX
```

```
FaritX = IdOuFun | CteLiteral | '(' Econc ')'
     | ‘OP_SIZE’ 'id'

OpRelac = ‘OP_MAIOR’ | ‘OP_MAIOR_IG’ | ‘OP_MENOR’
          |‘OP_MENOR_IG’ | ‘OP_IGUAL’ | ‘OP_N_IGUAL’

OpArit = ‘OP_ADD’ | ‘OP_SUB’

OpMult = ‘OP_MULT’ | ‘OP_DIV’

OpPot = ‘OP_POT’ | ‘OP_MOD’

CteLiteral = 'CONST_INT' | 'CONST_FLOAT' | 'CONST_CHAR'
               | 'CONST_STRING' | 'CONST_BOOL'

IdOuFun = 'id' ChamadaFunOrArr

ChamadaFunOrArr = '(' ParamsChamada ')' | '[' Earit ']' | ε
```